Práctica 3

Sistemas de ecuaciones diferenciales lineales

3.1. Introducción

Consideremos un sistema de ED lineales que involucra n funciones escalares $y_1(x), \ldots, y_n(x)$, dependientes de la variable x en su **forma normal:**

$$\begin{aligned} Dy_1 &= a_{11}(x)y_1 \cdots + a_{1n}(x)y_n + f_1(x) && (3.1)\\ &\vdots && (3.2)\\ Dy_n &= a_{n1}(x)y_1 \cdots + a_{nn}(x)y_n + f_n(x) && (3.3) \end{aligned}$$

Estas ecuaciones se pueden escribir en **forma matricial**

$$D\left\{\begin{array}{c} y_1 \\ \vdots \\ y_n \end{array}\right\} = \begin{pmatrix} a_{11} & \cdots & a_{1n} \\ \vdots & \ddots & \vdots \\ a_{n1} & \cdots & a_{nn} \end{pmatrix} \left\{\begin{array}{c} y_1 \\ \vdots \\ y_n \end{array}\right\} + \left\{\begin{array}{c} f_1 \\ \vdots \\ f_n \end{array}\right\} \tag{3.4}$$

y en forma compacta

$$DY = AY + F \tag{3.5}$$

Ejemplo 3.1.1 *Resolver el sistema de ecuaciones con el comando DSolve*

$$D\left\{\begin{array}{c} y_1 \\ y_2 \end{array}\right\} = \begin{pmatrix} 0 & 1 \\ -6 & -5 \end{pmatrix} \left\{\begin{array}{c} y_1 \\ y_2 \end{array}\right\} + \left\{\begin{array}{c} 0 \\ \cos(x) \end{array}\right\}, \qquad \left\{\begin{array}{c} y_1(0) \\ y_2(0) \end{array}\right\} = \left\{\begin{array}{c} 1 \\ 0 \end{array}\right\} \tag{3.6}$$

El DSolve se utiliza de manera similar a las ecuaciones diferenciales de una variable

```
In[]:= DSolve[{y1'[t]==y2[t], y2'[t]==-5y2[t]-6y1[t]+Cos[t],
       y1[0]==1,y2[0]==0}, {y1[t], y2[t]}, t]
```

obteniendo, como solución

$$\begin{aligned} Out[] \;\; := \;\; & \{\{y1[t] \to \tfrac{1}{10}e^{-3t}\left(-17+26e^{t}+e^{3t}Cos[t]+e^{3t}Sin[t]\right), \\ & y2[t] \to \tfrac{1}{10}e^{-3t}\left(51-52e^{t}+e^{3t}Cos[t]-e^{3t}Sin[t]\right)\}\} \end{aligned}$$

Si no se incluyen las condiciones iniciales nos deben de salir tantas constantes arbitrarias como ecuaciones (si el sistema estra escrito en su forma normal), que en este ejemplo serían $C[1]$ y $C[2]$.

En esta práctica resolveremos también estos sistemas de EDs sin utilizar el comando *DSolve*. En concreto consideraremos técnicas matriciales. Otra forma de proceder es el eliminar variables de manera similar a la resolución de ecuaciones lineales y resolver ecuaciones de una sola variable, pero en la que aparecen derivadas de órdenes superiores.

Relación con EDs lineales de orden superior Dada la ED lineal de orden n

$$y^{(n)} + a_{n-1}(x)y^{(n-1)} + \cdots + a_1(x)y' + a_0(x)y = f(x) \tag{3.7}$$

si hacemos los cambios de variables

$$y_1 = y, \quad y_2 = y' = Dy_1, \quad \ldots, \quad y_n = y^{(n-1)} = Dy_{n-1},$$

se puede reescribir como un sistema de ED lineales de primer orden

$$D\left\{\begin{array}{c} y_1 \\ \vdots \\ y_n \end{array}\right\} = \begin{pmatrix} 0 & 1 & \cdots & 0 \\ 0 & 0 & & \vdots \\ \vdots & & \ddots & 0 \\ & & & 1 \\ -a_0 & & \cdots & -a_{n-1} \end{pmatrix} \left\{\begin{array}{c} y_1 \\ \vdots \\ y_n \end{array}\right\} + \left\{\begin{array}{c} 0 \\ \vdots \\ f \end{array}\right\} \tag{3.8}$$

Ejemplo 3.1.2 *Resolver la ecuación diferencial de segundo orden como un sistema de dos ecuaciones*

$$y'' + 5\,y' + 6\,y = \cos(t), \qquad y(0) = 1, \quad y'(0) = 0$$

Si realizamos el cambio de variables

$$y_1 = y, \qquad y_2 = y'$$

se obtiene el sistema del ejemplo anterior. Si resolvemos la ecuación de segundo orden con el DSolve

```
In[]:= DSolve[{y''[t] + 5 y'[t] + 6y[t] ==  Cos[t],
           y[0] == 1, y'[0] == 0}, y[t], t]
```

obteniendo, como solución

$$Out[] := \{\{y[t] \to \frac{1}{10}e^{-3t}\left(-17+26e^{t}+e^{3t}Cos[t]+e^{3t}Sin[t]\right)\}\}$$

Vemos que se corresponde con $y_1(t)$ del ejemplo anterior. Además, podemos comprobar que $y_2(t)$ se corresponde con su derivada.

3.2. Método Matricial

3.2.1. Sistema lineal homogéneo

Diremos que el sistema de EDs lineal (3.6) es homogéneo si $F = 0$, esto es $DY = AY$. En tal caso, por ser el sistema lineal, si Y_1 y Y_2 son dos soluciones particulares de la ED (esto es $DY_i = AY_i,\ i = 1, 2$) entonces $Y_3 = \alpha Y_1 + \beta Y_2$, con α, β constantes arbitrarias, es tambén solución de la ED. Si el conjunto de funciones vectoriales $\{Y_1, \ldots, Y_n\}$ son tales que todas son soluciones particulares de la ED $DY = AY$ y además son linealmente independientes (LI), entonces forman un **sistema fundamental de soluciones**, y la matriz

$$\Phi = [Y_1\ \ldots\ Y_n]$$

se llama matriz fundamental de soluciones.

Teorema 3.2.1 *Si $\Phi = [Y_1\ \ldots\ Y_n]$ es la matriz fundamental de soluciones en $x \in (a, b)$ del sistema lineal homogéneo, $DY = AY$, de orden n, donde A es continua en $x \in (a, b)$, entonces su integral general es*

$$Y = C_1Y_1 + \ \ldots\ + C_nY_n$$

o equivalentemente

$$Y = \Phi C$$

con $C = (C_1, \ldots, C_n)^T$, siendo $C_i,\ i = 1, \ldots, n$ constantes arbitrarias.

Dadas n soluciones de la ED, $\{Y_1, \ldots, Y_n\}$, para saber si son o no LI nos construiremos el wronskiano $W(Y_1, \ldots, Y_n; x)$ dado por

$$W(Y_1, \ldots, Y_n; x) = \begin{vmatrix} y_{11} & \cdots & y_{1n} \\ \vdots & \ddots & \vdots \\ y_{n1} & \cdots & y_{nn} \end{vmatrix} \tag{3.9}$$

donde Y_j se corresponde con el vector columna, es decir $Y_j = (y_{1j}(x), \ldots, y_{nj})^T$. Entonces, $\{Y_1, \ldots, Y_n\}$, son LI si y sólo si $W(Y_1, \ldots, Y_n; x) \neq 0$.

Ejemplo 3.2.1 *Dado el sistema de ecuaciones diferenciales*

$$D \left\{ \begin{array}{c} x \\ y \\ z \end{array} \right\} = \begin{pmatrix} 4 & 0 & -2 \\ 0 & 4 & -4 \\ -2 & -4 & 5 \end{pmatrix} \left\{ \begin{array}{c} x \\ y \\ z \end{array} \right\} \tag{3.10}$$

comprueba si las siguientes funciones vectoriales son solución

$$y_1 = \left\{ \begin{array}{c} 1 \\ 2 \\ 2 \end{array} \right\}, \qquad y_2 = \left\{ \begin{array}{c} -2 \\ 1 \\ 0 \end{array} \right\} e^{4t}, \qquad y_3 = \left\{ \begin{array}{c} 1 \\ 2 \\ -\frac{5}{2} \end{array} \right\} e^{9t}.$$

En caso afirmativo, comprobar si son linealmente independientes calculando el wronskiano correspondiente

En primer lugar, definimos la matriz y los vectores

```
In[]:= m = {{4, 0, -2}, {0, 4, -4}, {-2, -4, 5}}
       y1 = {{1}, {2}, {2}};
       y2 = {{-2}, {1}, {0}}E^(4t);
       y3 = {{1}, {2},{-5/2}}E^(9t);
```

Podemos comprobar la forma matricial de éstos con el comando Matrix-Form[m] (para asegurarnos que no escribimos, por ejemplo, la matriz traspuesta). A continuación, recordando que el producto de matrices se realiza en Mathematica con un punto vemos que

```
In[]:= D[y1, t] - m.y1
       D[y2, t] - m.y2
       D[y3, t] - m.y3
```

dándonos tres vectores nulos, por lo que son solución. Con el comando Join nos podemos construir la matriz fundamental de soluciones (el comando sirve para unir matrices o vectores uno encima del otro, por lo que hay que utilizarlo con el comando Transpose)

```
In[]:=phi=Transpose[Join[Transpose[y1],Transpose[y2],Transpose[y3]]];
     Det[phi]
```

dándonos

$$Out[] := -\frac{45\ e^{13t}}{2}$$

y como no se anula nunca, son soluciones independientes

Nuestro objetivo será buscar n soluciones independientes del sistema de ED. Similarmente a como se hacía en la ED lineales de órdenes superiores, buscaremos soluciones exponenciales. Dada la función vectorial

$$Y = \left\{ \begin{array}{c} \alpha_1 \\ \vdots \\ \alpha_n \end{array} \right\} e^{\lambda x},$$

veamos qué condiciones tiene que satisfacer λ y el vector de constantes $\bar{\alpha} = (\alpha_1, \ldots, \alpha_n)^T$ para que sea solución de la ecuación. Si derivamos la función vectorial Y vemos inmediatamente que

$$DY = \lambda Y$$

y como la ecuación es $DY = AY$, igualando las partes derechas de las ecuaciones tenemos

$$\begin{pmatrix} a_{11} & \cdots & a_{1n} \\ \vdots & \ddots & \vdots \\ a_{n1} & \cdots & a_{nn} \end{pmatrix} \left\{ \begin{array}{c} \alpha_1 \\ \vdots \\ \alpha_n \end{array} \right\} e^{\lambda x} = \lambda \left\{ \begin{array}{c} \alpha_1 \\ \vdots \\ \alpha_n \end{array} \right\} e^{\lambda x} \tag{3.11}$$

y eliminando $e^{\lambda x}$ nos queda un problema de valores propios, donde λ es el valor propio y $\bar{\alpha} = (\alpha_1, \ldots, \alpha_n)^T$ es el vector propio asociado.

Teorema 3.2.2 *Si A es una matriz constante y $\sigma(A)$ está formada por n números (reales o complejos), $\lambda_1, \ldots, \lambda_n$, distintos entre sí, entonces un sistema fundamental de soluciones del sistema $DY = AY$ esta formado por $\{\bar{\alpha}_1 e^{\lambda_1 x}, \ldots, \bar{\alpha}_1 e^{\lambda_n x}\}$, donde $\bar{\alpha}_i$ es un autovector (no nulo) asociado al autovector λ_i, $i = 1, \ldots, n$.*

Ejemplo 3.2.2 *Halla los valores y vectores propios de la matriz (3.10)*

Podemos calcular los valores y vectores propios de la matriz m con los comandos

```
In[]:=Eigenvalues[m]
      Eigenvectors[m]
```

dándonos

```
Out[]:= {9, 4, 0}
        {{-2, -4, 5}, {-2, 1, 0}, {1, 2, 2}}
```

que se corresponden con los resultados para y_1, y_2, y_3 *. El Mathematica dispone también de un comando que nos da los valores propios y sus vectores propios correspondientes en un solo comando*

```
In[]:=Eigensystem[m]
```

dando

```
Out[]:= {{9, 4, 0}, {{-2, -4, 5}, {-2, 1, 0}, {1, 2, 2}}}
```

Valores y vectores propios complejos Si $\lambda_1 = \delta_1 + i\delta_2$ es un valor propio complejo con vector propio, también complejo, $\alpha_1 = \gamma_1 + i\gamma_2$ con $\gamma_1, \gamma_2 \in \mathbb{R}^n$ entonces su conjugado $\lambda_2 = \delta_1 - i\delta_2$ es también un valor propio con vector propio $\alpha_2 = \gamma_1 - i\gamma_2$. Al igual que ocurría en el capítulo anterior, la combinación de estas dos soluciones tiene que ser tal que al final el resultado sea real. En la combinación

$$C_1(\gamma_1 + i\gamma_2)e^{(\delta_1+i\delta_2)x} + C_2(\gamma_1 - i\gamma_2)e^{(\delta_1-i\delta_2)x}$$

las constantes complejas, C_1, C_2, tienen que ser una la compleja conjugada de la otra para que de una función vectorial real. Si se desarrlla (revisar la teoría) se tiene que la contribución de las dos soluciones será

$$y = \Big(D_1(\gamma_1 \cos(\delta_2 x) - \gamma_2 \sin(\delta_2 x)) + D_2(\gamma_1 \sin(\delta_2 x) + \gamma_2 \cos(\delta_2 x))\Big)e^{\delta_1 x} \quad (3.12)$$

con D_1, D_2 dos constantes reales.

Ejemplo 3.2.3 *Consideremos un sistema de dos muelles acoplados y con rozamiento. El desplazamiento respecto de la posición de equilibrio de ambos muelles viene descrito por el siguiente sistema de ecuaciones diferenciales acopladas*

$$\begin{aligned} m_1 \frac{d^2 x_1}{dt^2} &= -b_1 \frac{dx_1}{dt} - k_1 x_1 + k_2(x_2 - x_1) + b_2 \left(\frac{dx_2}{dt} - \frac{dx_1}{dt}\right) \\ m_2 \frac{d^2 x_2}{dt^2} &= -k_2(x_2 - x_1) - b_2 \left(\frac{dx_2}{dt} - \frac{dx_1}{dt}\right) \end{aligned}$$

donde m_i *son las masas,* k_1 *es la constante de restitución del cuerpo* m_1 *que esta sujeto a un punto fijo y* k_2 *es la constante de restitución del cuerpo* m_2

que esta sujeto al cuerpo m_1, y b_i los coeficientes de rozamiento. Este sistema se puede reescribir como un sistema de ecuaciones de primer orden con los siguientes cambios

$$y_1 = x_1, \quad y_2 = \frac{dx_1}{dt}, \quad y_3 = x_2, \quad y_4 = \frac{dx_2}{dt}$$

con lo que se obtiene la ecuación matricial

$$D \left\{ \begin{array}{c} y_1 \\ y_2 \\ y_3 \\ y_3 \end{array} \right\} = \left(\begin{array}{cccc} 0 & 1 & 0 & 0 \\ -\left(\frac{k_1}{m_1} + \frac{k_2}{m_2}\right) & -\left(\frac{b_1}{m_1} + \frac{b_2}{m_2}\right) & \frac{k_2}{m_1} & \frac{b_2}{m_1} \\ 0 & 0 & 0 & 1 \\ \frac{k_2}{m_2} & \frac{b_2}{m_2} & -\frac{k_2}{m_2} & -\frac{b_2}{m_1} \end{array} \right) \left\{ \begin{array}{c} y_1 \\ y_2 \\ y_3 \\ y_3 \end{array} \right\} \tag{3.13}$$

Tomamos los valores

$$m_1 = 10, \quad m_2 = 20, \quad k_1 = 5, \quad k_2 = 7, \quad b_1 = 1, \quad b_2 = 1$$

y las condiciones iniciales

$$y_1 = 1, \quad y_2 = 0, \quad y_3 = -1, \quad y_4 = 0$$

Dibujar la solución $x_1(t), x_2(t)$ para $t \in [0, 100]$. Comprobar que, aunque el movimiento de ambos muelles inicialmente es independiente, terminan sincronizados.

Solución

```
In[]:=m1 = 10.0; m2 = 20.0; k1 = 5.0; k2 = 7.0; b1 = 1.0; b2 = 1.0;
      ma2 = {{0,1,0,0},{-(k1/m1+k2/m2),-(b1/m1+b2/m2),k2/m1,b2/m1},
           {0,0,0,1},{k2/m2,b2/m2,-k2/m2,-b2/m2}};
      MatrixForm[ma2]
```

A continuación calculamos los valores y vectores propios de ma2

```
In[]:= {l1, l2, l3, l4} = Eigenvalues[ma2];
       {v1, v2, v3, v4} = Eigenvectors[ma2];
```

Como los valores y vectores propios son complejos, tomaremos las partes reales e imaginarias de ellos y aplicaremos (3.12)

```
In[]:= l1r = Re[l1]; l1i = Im[l1]; l3r = Re[l3]; l3i = Im[l3];
       v1r = Re[v1]; v1i = Im[v1]; v3r = Re[v3]; v3i = Im[v3];
```

```
In[]:= {y1[x_], y2[x_], y3[x_],y4[x_]} =
        (C1(v1r*Cos[l1i*x] - v1i*Sin[l1i*x]) +
         C2(v1r*Sin[l1i*x] + v1i*Cos[l1i*x]))E^(l1r*x) +
        (C3(v3r*Cos[l3i*x] - v3i*Sin[l3i*x]) +
         C4(v3r*Sin[l3i*x] + v3i*Cos[l3i*x]))E^(l3r*x);
```

```
In[]:=Solve[{y1[0]==1,y2[0]==0,y3[0]==-1,y4[0]==0},{C1,C2,C3,C4}]
```

```
Out[]:={{C1->0.0160692,C2->2.18049,C3->0.558043,C4->-0.0631368}}
```

y sustituyendo se obtiene la solución. Teniendo en cuenta que x_1, x_2 *se corresponden son* y_1, y_3 *podemos dibujar las soluciones para observar sus evoluciones*

```
In[]:= Plot[{y1[x], y3[x]}, {x, 0, 100}]
```

Matrices no invertibles Si un valor propio λ_i tiene multiplicidad m y tenemos m vectores propios LI, $\bar{\alpha}_{i1}, \ldots, \bar{\alpha}_{im}$ (esto ocurre si la matriz A es diagonalizable) entonces m soluciones LI serán $\bar{\alpha}_{i1}e^{\lambda_i x}, \ldots, \bar{\alpha}_{im}e^{\lambda_i x}$.

Si la matriz A no es diagonalizable, entonces encontraremos menos vectores propios que la multiplicidad del valor propio. En tal caso, probaremos con las siguientes funciones vectoriales

$$\left\{ \begin{array}{c} \alpha_1 x + \beta_1 \\ \vdots \\ \alpha_n x + \beta_n \end{array} \right\} e^{\lambda_i x}, \qquad \left\{ \begin{array}{c} \gamma_1 x^2 + \delta_1 x + \epsilon_1 \\ \vdots \\ \gamma_n x^2 + \delta_n x + \epsilon_n \end{array} \right\} e^{\lambda_i x}, \quad \text{etc.}$$

Ejemplo 3.2.4 *Resolver el siguiente sistema sin utilizar DSolve*

$$\left\{ \begin{array}{l} y_1' = -8y_1 - y_2 \\ y_2' = 16y_1 \end{array} \right. \tag{3.14}$$

Comprobar el resultado con el DSolve.

Sol.: *En primer lugar definimos la matriz asociada y calculamos sus valores y vectores propios*

```
In[]:=m3={{-8, -1}, {16, 0}}
      Eigensystem[m3]
```

dándonos

```
Out[]:= {{-4, -4}, {{-1, 4}, {0, 0}}}
```

esto es, un valor propio -4 doble, pero solo un vector propio, por lo que la matriz no es invertible. Para obtener el segundo vector solución probaremos

$$v_2 = \left\{ \begin{array}{c} \alpha_1 x + \beta_1 \\ \alpha_2 x + \beta_2 \end{array} \right\} e^{-4x},$$

y sustituimos en la ecuación diferencial

```
In[]:=v1 = {-1, 4};
      v2 = {{al1*x + be1}, {al2*x + be2}}*E^(-4x);
      D[v2, x] - m3.v2 // Simplify
```

dando

$$Out[] := \{\{e^{-4\,x}(4be1{+}be2{+}al1{+}4al1\;x{+}la2\;x), \{e^{-4\,x}(-16be1{-}4be2{+}al2{-}16al1\;x{-}4al2\;x))\}\}$$

Para que se anule se debe cumplir que

```
In[]:=Solve[{4be1+be2+al1==0, 4al1+al2==0,-16be1-4be2+al2==0,
          -16al1-4al2==0}, {al1,be1,al2,be2}]
```

dando

$$Out[] := \{\{be1 \to -\frac{be2}{4} + \frac{al2}{16}, \quad al1 \to -\frac{al2}{4}\}\}$$

por lo que tenemos libertad en la elección de $al2$ y $be2$. Si tomamos los valores $be2 = 4,\ al2 = 16$ llegaremos a la solución con el comando

```
In[]:=ys[x_] = c1*v1*E^(-4x) + c2*v2
```

Con el comando DSolve obtenemos una solución aparentemente distinta pero que en realidad es equivalente debido a que las constantes estan distribuidas de manera diferente.

3.2.2. Sistema lineal no homogéneo

Dado el sistema no homogéneo

$$DY = AY + F$$

la solución general viene dada por la suma de la solución de la homogénea $Y_H = C_1 Y_1 + \cdots + C_n Y_n$, y un asolución particular de la no homogénea, Y_p, esto es $Y = Y_H + Y_p$. Esto es así porque tiene n constantes independientes y satisface la ED no homogénea

$$\begin{aligned} DY &= D(Y_H + Y_p) = C_1 DY_1 + \cdots + C_n DY_n + DY_p \\ &= C_1 AY_1 + \cdots + C_n AY_n + AY_p + F = AY + F. \end{aligned}$$

Por tanto, sólo nos falta buscar una solución particular de la no homogénea. Para ello utilizaremos **variación de constantes**, donde tomamos como solución particular la homogénea pero cambiando las constantes por funciones a determinar

$$Y_p = c_1(x)Y_1 + \cdots c_n(x)Y_n.$$

A las $c(x)$ se le imponen unas condiciones similares a las impuestas en el capítulo anterior para las ED lineales de orden superior

Teorema 3.2.3 *Sea $DY = AY + F$ un sistema lineal no homogéneo y $\{Y_1, \dots, Y_n\}$ un sistema fundamental de soluciones del sistema homogéneo asociado. entonces $Y_p = c_1(x)Y_1 + \cdots c_n(x)Y_n$ es una solución particular del sistema no homogéneo siempre que*

$$c_1' Y_1 + \cdots c_n' Y_n = F \tag{3.15}$$

Ejemplo 3.2.5 *Resolver el siguiente sistema sin utilizar DSolve*

$$\begin{cases} x' = 6x - 8y + e^{3t} \\ y' = -x \end{cases} \tag{3.16}$$

Comprobar el resultado con el DSolve.

Sol.: *En primer lugar definimos la matriz asociada y calculamos sus valores y vectores propios*

```
In[]:=mat2 = {{6, -8}, {1, 0}};
```

y calculamos lo svalores y vectores propios

```
In[]:={l1, l2} = Eigenvalues[mat2];
      {v1, v2} = Eigenvectors[mat2];
```

con lo que nos podemos construir la solución de la homogénea

```
In[]:=solH[t_] = C1*v1*E^(l1*t) + C2*v2*E^(l2*t);
```

A partir de aquí, resolvemos el sistema de ecuaciones para las c_i' dadas en (3.15).

```
In[]:=Solve[solH[t] == {E^(3t), 0}, {C1, C2}];
```

obteniendo

$$Out[] := \{\{C1 \to \frac{e^{-t}}{2},\ C2 \to -\frac{e^{t}}{2}\}\}$$

Calculamos las integrales de estas funciones (añadiendo las constantes $K1, K2$)

```
In[]:= dCs[t_] = { E^(-t)/2, -E^(t)/2};
       {C1, C2} = Integrate[dCs[t], t] + {K1, K2}
```

Con todo esto, la solución viene dada por

```
In[]:= solH[t]
```

Con el comando DSolve obtenemos una solución aparentemente distinta pero, al igual que en ejemplos anteriores, es equivalente debido a que las constantes estan distribuidas de manera diferente.